


## Introductory Operating Guide for the HP 3326A Two-Channel Synthesizer with the HP 9000 Series 200 Desktop Computer (BASIC)



### INTRODUCTION

This programming note is an introduction to the remote operation of the HP 3326A Two Channel Synthesizer. System setup and checkout instructions are included, along with a number of example programs. These programs will demonstrate some of the capabilities of the 3326A and can serve as a starting point for programs tailored to a user's specific needs.

The HP 3326A Two Channel Synthesizer combines two independent synthesizers, flexible modulation, and control circuitry into a single instrument. It provides precise phase-offset, two-tone sweep, random frequency switching, internal modulation, and pulse signals for bench or systems use.

All examples demonstrate HP-IB control of the 3326A using the HP 9836A Desktop Computer and the enhanced BASIC 2.0 programming language. While the 9836A is referenced in all program examples, these programs will also run on the other HP 9000 Series 200 desktop computers such as the Model 216 (9816A) and Model 226 (9826A).

The topics to be covered through the use of example programs include:

- REMOTE vs. LOCAL operation
- Basic parameter setup of both channels
- Sweep operations including continuous, stepped, and discrete
- Instrument state storage and retrieval
- Service requests and error retrieval



Also included are appendices on programming codes, error codes, and the status byte of the 3326A.

## REFERENCE INFORMATION

For further information on the HP Interface Bus:

- Tutorial Description of the HP-IB  
HP Lit. No. 5952-0156
- Condensed Description of the HP-IB  
HP Part No. 59401-90030

For further information on the HP 3326A:

- HP 3326A Operating Manual  
HP Part No. 03326-90000
- Product Note 3326A-1 Quick Reference Guide to the HP 3326A  
HP Lit. No. 5953-5134

For further information on the HP 9836A:

- Operating Manual  
HP Part No. 09836-90000
- BASIC Language Reference  
HP Part No. 09826-90056
- BASIC 2.0 Condensed Reference  
HP Part No. 09826-90051
- BASIC Programming Techniques  
HP Part No. 09826-90011
- BASIC Interfacing Techniques  
HP Part No. 09826-90025

## EQUIPMENT REQUIRED

To perform the examples described in this programming note, you will need the following equipment and accessories:

- HP 3326A Two Channel Synthesizer
- HP 9836A Desktop Computer with BASIC Programming Language (option 011 or 711)
- HP 10833A HP-IB Cable (or equivalent)

### NOTE

*The following equipment is not required for the programs to function but rather for a visual display of the 3326A output*

- HP 1740A or equivalent two-channel oscilloscope
- Two BNC cables (HP 11170C or equivalent)

## EQUIPMENT SETUP

Begin by removing power from the 3326A and 9836A.

1. As shown in Figure 1, use the 10833 HP-IB cable to connect the 3326A to the built-in HP-IB interface of the 9836A.




Figure 1. System HP-IB Connection

### CAUTION

*Do not attempt to mate black metric threaded screws on one connector with silver English threaded nuts on another connector, or vice versa, as damage may result. A metric conversion kit, which will convert one cable and one or two instruments to metric hardware is available by ordering HP Part No. 5060-0138.*

2. If a visual display of the 3326A outputs is desired, connect the 3326A channel A and B outputs to the two inputs of a two-channel oscilloscope using two BNC cables. The 3326A outputs should both be terminated in 50 ohms.
3. If required, load the BASIC language operating system, following the instructions in the BASIC Operating Manual for the HP 9836A. (9836A option 711 only)
4. Turn power ON to the 3326A. All programs in this note assume the 3326A HP-IB address is decimal 18, as preset at the factory.



To display the current HP-IB address of the 3326A, press the front-panel SHIFT key, followed by the LOCAL key. The address will appear in the 3326A display as:

A d d r . = XX

where XX is a number from 00 to 30, inclusive.

If the address is not correct, enter the correct address (18) with the numeric key pad. The address is set and displayed when the second digit is entered.

The nonvolatile memory of the 3326A will retain the address while the instrument is on or off, until another address is entered.

## CHECKOUT

If the BASIC operating system has been properly loaded, "BASIC READY" should appear in the 9836A display (option 711 only). If this does not occur, reload BASIC.

The following steps verify that the HP-IB connections and interface are functional:

1. Press the CLR I/O key of the 9836A to eliminate any possibility of a bus hangup.
2. Press both the SHIFT and PAUSE keys of the 9836A at the same time to reset the computer.
3. Type in "SCRATCH" on the 9836A keyboard and press the EXECUTE key. This clears any previous programs from memory.
4. Type in "REMOTE 7" on the 9836A keyboard and press the EXECUTE key. On the 3326A, both the green REMOTE and yellow LISTEN indicators should be lit.
5. If these two annunciators are not lit, perform the EQUIPMENT SETUP procedure once again and repeat this CHECKOUT procedure.

If the CHECKOUT procedure fails a second time, look for instrument or controller errors and consult the appropriate manual as listed in the REFERENCE INFORMATION section of this document.

6. When this checkout procedure passes successfully, type in "LOCAL 7" on the 9836A keyboard, followed by the EXECUTE key. The HP-IB connections and interface are now functional and the programming examples can be performed.

## PROGRAMMING EXAMPLES

The following example programs cover a number of basic and advanced topics important in programming the HP 3326A.

Each programming example uses the following format:

- A general description of the program and its purpose
- A program listing
- Instructions for running the program and an explanation of the various sections of the program

### EXAMPLE PROGRAM NO. 1

#### REMOTE, LOCAL, and LOCAL LOCKOUT operation

When operated from its front panel, the 3326A is in the LOCAL mode of operation. All front panel controls are active.

When the 3326A is under program control on the HP-IB (IEEE-488) bus, LOCAL control is disabled and the front panel is inactive. This is the REMOTE mode of operation. In this mode, the instrument can be restored to front panel control by pressing the LOCAL key or sending a LOCAL command on the HP-IB.

Issuing the LOCAL LOCKOUT command prevents all LOCAL operation and disables the action of the LOCAL key, along with the rest of the 3326A front panel controls. Front panel control can only be restored to the instrument by a LOCAL command from the controller.

The following program demonstrates these modes of operation and shows how a variable can be used in a program to define the address of a particular instrument.

1. RESET the 9836A
2. Type in SCRATCH and press the EXECUTE key to clear any previous programs.
3. Press the EDIT key, then the EXECUTE key, and type in the following program:

```
10  ! REMOTE, LOCAL, LOCAL LOCKOUT DEMO
20  !
30  Source=718
40  !
50  REMOTE Source
60  !
70  DISP "3326A IS IN REMOTE MODE"
80  !
90  PAUSE
100 REMOTE Source
110 LOCAL LOCKOUT 7
120 DISP "3326A FRONT PANEL LOCKED OUT"
130 !
140 PAUSE
150 LOCAL Source
160 DISP "3326A FRONT PANEL ACTIVE"
170 !
180 PAUSE
190 OUTPUT Source;"RST"
200 LOCAL Source
210 END
```



## PROGRAM 1—EXPLANATION AND OPERATION

After the program has been entered, RESET the 9836A and press the RUN key to begin execution of the program. No oscilloscope is necessary.

Instead of using the actual instrument address (718), **line 30** assigns the address to a variable that can then be used as a device "name". This makes the program easier to understand and if the instrument address changes, only one line of the program need be changed.

**Line 50** places the 3326A in the REMOTE mode, and **line 70** displays a message to the user on the 9836A CRT. **Line 90** pauses execution of the program.

At this point, both the green REMOTE and yellow LISTEN indicators on the 3326A should be lit. Attempt to modify an instrument parameter such as FREQUENCY or AMPLITUDE. Verify that all keys except LOCAL are disabled and that an error message is displayed. Press the LOCAL key. The REMOTE annunciator will go out and the 3326A can be operated normally from the front panel.

Press CONTINUE on the 9836A to resume execution of the program. **Line 110** places all instruments on the bus in the LOCAL LOCKOUT mode and program execution is paused.

Now verify that the 3326A is again in the REMOTE mode. This time, however, the LOCAL key will not return the instrument to front panel control and will generate an error message. The LOCAL LOCKOUT mode is very useful for preventing unwanted changes in parameters or states.

Press CONTINUE on the 9836A to resume execution of the program. **Line 150** returns all instruments on the bus to front panel control and program execution is halted. Verify that the 3326A front panel controls are active and that the REMOTE annunciator is extinguished.

CONTINUE the program. In **Line 190** an OUTPUT statement is used to command the 3326A to perform an INSTRUMENT PRESET (RST). The OUTPUT statement is a very common one in most programs and serves two functions—it first places the instrument in the REMOTE mode and then passes information or commands. In **line 200** the 3326A is returned to LOCAL control.

### EXAMPLE PROGRAM NO. 2 BASIC PARAMETER ENTRY

The majority of programming for the 3326A usually involves setting the basic operational parameters such as mode, channel, frequency, amplitude, etc. This program demon-

strates several ways of setting up these parameters with data from the program itself and data entered by the user. This program also covers other topics such as autocalibration and the best order for parameter entry.

RESET the 9836A, SCRATCH the memory, and press EDIT and EXECUTE to enter the following program:

```
10      ! BASIC PARAMETER ENTRY
20      !
30      Source=718
40      !
50      OUTPUT Source;"RST, CAL"
60      !
70      !CHANNEL A PARAMETER ENTRY:
80      OUTPUT Source;"CHA"
90      OUTPUT Source;"FCNA SQR, FR 1.2345 KHZ"
100     OUTPUT Source;"AM 2 VQ, OF +1.05 VQ"
110     !
120     !CHANNEL B PARAMETER ENTRY:
130     OUTPUT Source;"CHB"
140     OUTPUT Source;"FCNB SIN, FR 9876 HZ"
150     OUTPUT Source;"AM 1.05 VRMS; OF -0.8 VQ"
160     LOCAL Source
170     PAUSE
180     !
190     !SET UP FOR INTERNAL MODULATION
200     OUTPUT Source;"RST"
210     OUTPUT Source;"CHA, FR 100 KHZ, AM 10 DBM"
220     OUTPUT Source;"CHB, FR 4.0 KHZ"
230     OUTPUT Source;"AIA ON, CAL"
240     OUTPUT Source;"ML 0 PC"
250     !
260     !INTERROGATE USER FOR INDEX
270     PRINT CHR$(12)
280     INPUT "ENTER MODULATION INDEX IN PERCENT",Index
290     IMAGE "MODULATION LEVEL IS ",DDD.DD," PERCENT"
300     PRINT USING 290;Index
310     IMAGE "ML ",DDD.DD," PC"
320     OUTPUT Source USING 310;Index
330     GOTO 280
340     END
```

## PROGRAM 2—EXPLANATION AND OPERATION

After the program has been entered, RESET the 9836A and press the RUN key to begin execution of the program.

Set up the oscilloscope with channel 1 and 2 sensitivity of 1 Volt/div. and 0.2 ms/div.

**Lines 50** initializes the 3326A. The RST command performs an instrument preset, leaving the 3326A in a known state. This insures that no previously used parameters or states will be incompatible with what is about to be programmed. A calibration is then performed.

In the preset mode that results from the RST command, autocalibration is disabled. If the 3326A is in the AUTOCAL ON mode, the periodic calibrations that are automatically performed may interrupt program execution. By turning these calibrations off and performing them only when necessary, such interruptions are prevented.

**Lines 80-100** set up channel A. As parameters are entered from the front panel or under program control, the intermediate states that result must not create errors. These errors will cause the improper parameters to be rejected. To guard



against errors, the 3326A is first preset (RST) and parameters are entered in the following order:

- Instrument Mode
- Channel
- Output Function
- Frequency
- Amplitude
- Misc. Parameters:
  - DC Offset
  - Phase
  - Sweep Frequencies/Times
  - Modulation

**Lines 130-150** set up channel B in the same manner as channel A. Several parameters may be sent with a single output statement; simply separate them with a semicolon, comma, or space.

**Line 160** returns the 3326A to LOCAL control so the instrument state can be examined from the front panel. This is a chance to select a channel and press one of the ENTRY block keys to verify that the 3326A is set up according to **lines 90, 100, 140, and 150** of the program.

Press CONTINUE on the 9836A to resume program execution. **Lines 210-220** set the 3326A up for internal modulation with a carrier frequency of 100 kHz and modulation frequency of 4 kHz. **Line 230** turns internal modulation on and performs a calibration. Modulation level is preset at 0% in **line 240**.

**Line 270** clears the 9836A display by printing a form feed. **Line 280** requests that the user input the modulation index in percent and pauses for the data to be entered.

Enter a modulation index from 0.00 percent to 100.00 percent and press CONTINUE.

**Lines 290-300** print the desired modulation index on the CRT. The IMAGE statement is used to format the output and round it to two digits to the right of the decimal point.

**Lines 310-320** use a similar IMAGE statement to OUTPUT the desired modulation index to the 3326A. The GOTO statement in **line 330** repeats the section of the program that enables the user to enter a new modulation index.

### EXAMPLE PROGRAM NO. 3 SWEEP PROGRAMMING

The 3326A can perform a variety of frequency sweeps, with one or both channels. These sweeps are programmed in much the same way as they would be set up from the front panel.

This program demonstrates sweeps in the Two Channel and Two-Phase modes, with two different types of sweep triggering.

RESET the 9836A, SCRATCH the memory, and press EDIT and EXECUTE to enter the following program:

```

10  ! SWEEP PROGRAMMING
20  !
30  Source=718
40  !
50  OUTPUT Source;"RST, CAL"
60  !
70  ! SET UP FOR TWO-CHANNEL SWEEP
80  OUTPUT Source;"CHA; AM 3.5 V0"
90  OUTPUT Source;"ST 750 HZ; SP 12.5 KHZ; STIM 1.25 SEC"
100 !
110 OUTPUT Source;"CHB"
120 OUTPUT Source;"FCNB SQR; AM 3 V0"
130 OUTPUT Source;"ST 12.5 KHZ; SP 750 HZ"
140 OUTPUT Source;"SC"
150 DISP "CONTINUOUS, TWO-CHANNEL SWEEP"
160 PAUSE
170 !
180 ! SET UP FOR TWO-PHASE SWEEP
190 Stop_freq=1000
200 OUTPUT Source;"RST"
210 OUTPUT Source;"MODE TWOP"
220 OUTPUT Source;"CHA, AM 0 DBU"
230 OUTPUT Source;"CHB, AM 0 DBU, PH -90 DEG"
240 !
250 OUTPUT Source;"SM TRGL, STIM 1.5 SEC"
260 OUTPUT Source;"ST 500 HZ, SP",Stop_freq,"HZ"
270 OUTPUT Source;"STS"
280 DISP "TRIGGER SINGLE TWO-PHASE SWEEP"
290 FOR I=1 TO 5
300 TRIGGER Source
310 WAIT 1.5
320 TRIGGER Source
330 WAIT 1.5
340 Stop_freq=Stop_freq*2
350 OUTPUT Source;"SP",Stop_freq,"HZ"
360 NEXT I
370 LOCAL Source
380 END

```

### PROGRAM 3—EXPLANATION AND OPERATION

After the program has been entered, RESET the 9836A and press the RUN key to begin execution of the program.

Set up the oscilloscope with channel 1 and 2 sensitivity of 1 Volt/div. and 0.2 ms/div. Experiment with different types of triggering.

**Line 90** sets up the 3326A for a sine wave sweep on channel A from 750 Hz to 12.5 kHz with a sweep time of 1.25 seconds. The default "ramp" sweep is used, where the instrument sweeps from start to stop frequency in the specified time and resets as quickly as possible for the next sweep. **Lines 120-130** configure channel B for a square wave sweep over the same limits as channel A. Sweep limits are reversed, however, to yield a downward sweep instead of an upward one.

**Line 140** initiates a continuous sweep on both channels and program execution is paused. It is best to view this sweep in the oscilloscope's ALT channel mode with independent triggers for each channel. As an alternative, the channels can be viewed separately.



CONTINUE the program to set up the next sweep. Setup for the Two-Phase sweep begins on **line 190** where an initial value is established for the stop frequency. In **lines 210-230**, the 3326A is placed in the Two-Phase mode with equal amplitudes on both channels and a  $-90$  degree phase offset for the channel B output. In **line 250** the sweep time is set at 1.5 seconds and the sweep mode is "triangle", where sweeps are from the start frequency to the stop frequency and back at equal rates.

**Line 260** sets the start frequency at 500 Hz and the stop frequency at 1000 Hz, the current value of the variable Stop-freq. This value will be modified by the program later for successive sweeps. In **line 270** the trigger mode of the 3326A is established as "Start Single", where both hardware and software trigger commands will result in single sweeps. In this case, triggers are provided by the selective device trigger command "TRIGGER Source" on **lines 300 and 320**. An alternative is the group execute trigger command "TRIGGER 7". The group execute trigger command will send a simultaneous trigger to all devices on the bus and may be used to start several events at once.

**Lines 290-360** form a loop that will repeatedly trigger sweeps in both directions and modify the stop frequency when each sweep is completed. The first trigger command causes the source to sweep upward to the stop frequency and stop. The program pauses for an appropriate interval to allow the sweep to be completed and sends a second trigger command, to cause the 3326A to sweep back to the start frequency.

When each sweep is concluded, **lines 340-350** cause the sweep stop frequency to be doubled and a new sweep is initiated. This process is repeated 5 times, once for each execution of the loop. **Line 370** returns the 3326A (and any other instruments on the bus) to LOCAL control.

#### EXAMPLE PROGRAM NO. 4 MODIFYING ENTRY PARAMETERS

Once the operating state of the 3326A has been set up, it is often necessary to modify one or more of the entry parameters (frequency, amplitude, phase, etc.). This can be done by sending successive new values for these parameters or by using the Entry Increment function.

The Entry Increment function (EINC) sends the 3326A a specific value which will then be used to increment or decrement the currently displayed entry parameter. Once the entry increment value has been sent, it is only necessary to send an up (UP) or down (DN) command to change the displayed parameter.

This technique has two major advantages: First, it is a very convenient way to make repetitive, changes of equal size. Second, it is considerably faster to send a succession of UP or DN commands than to reprogram specific values.

This program demonstrates the EINC function with frequency, amplitude, and phase changes.

RESET the 9836A, SCRATCH the memory, and press EDIT and EXECUTE to enter the following program:

```

10      ! MODIFYING ENTRY PARAMETERS
20      !
30      Source=718
40      !
50      OUTPUT Source;"RST, CAL"
60      !
70      OUTPUT Source;"CHA AM 4 V0, FR, EINC 250 HZ"
80      !
90      ! PERFORM STEPPED FREQ SWEEP
100     FOR I=1 TO 99
110         OUTPUT Source;"UP"
120         WAIT .1
130     NEXT I
140     PAUSE
150     !
160     OUTPUT Source;"CHA FR 15 KHZ, AM 1.5 V0, EINC 0.1 V0"
170     !
180     ! PERFORM STEPPED AMPL SWEEP
190     FOR I=0 TO 3
200         FOR J=0 TO 25
210             OUTPUT Source;"UP"
220             WAIT .1
230         NEXT J
240         FOR J=0 TO 25
250             OUTPUT Source;"DN"
260             WAIT .1
270         NEXT J
280     NEXT I
290     PAUSE
300     !
310     ! SET UP FOR TWO-PHASE OPERATION
320     OUTPUT Source;"RST, MODE TWOP"
330     OUTPUT Source;"CHA FR 2.5 KHZ, AM 4 V0"
340     OUTPUT Source;"CHB AM 4 V0"
350     !
360     ! SET ENTRY INCREMENT FOR PHASE
370     OUTPUT Source;"CHB, PH, EINC 30 DEG"
380     !
390     ! PERFORM PHASE INCREMENT
400     FOR I=1 TO 100
410         OUTPUT Source;"DN"
420         WAIT .2
430     NEXT I
440     END

```

#### PROGRAM 4—EXPLANATION AND OPERATION

After the program has been entered, RESET the 9836A and press the RUN key to begin execution of the program.

Set up the oscilloscope with channel 1 and 2 sensitivity of 1 Volt/div. and 0.2 ms/div.

**Line 70** sets channel A amplitude and sends a frequency entry increment of 250 Hz. When an entry increment is sent, it must be consistent with the current entry mode of the instrument.

To perform the frequency increment or step, only an UP or DN command is required. **Lines 100-130** set up a loop to send the UP command 99 times, with a delay of 0.1 second between commands. The delay slows program execution



enough to allow the 3326A to display the updated frequency each time a command is received. The PAUSE instruction then halts program execution.

Note that the use of EINC allows any frequency increment to be performed rapidly with only a two-letter command.

CONTINUE program execution. **Line 160** sets up the 3326A for a 15 kHz, 1.5 Volt sine wave output and an amplitude entry increment of 100 mV. **Lines 190-280** create two loops nested within a third. The inner loops step the amplitude up and then down in 25 steps, thus performing an amplitude sweep. The outer loop causes this sweep to be performed 4 times. The program is once again PAUSED.

Press CONTINUE to resume program execution. **Lines 320-340** set the instrument to the TWO PHASE mode with a frequency of 2.5 kHz and an amplitude of 4 Volts on each channel. **Line 370** sets a channel B phase entry increment of 30 degrees.

**Lines 400-430** decrement channel B phase in 100 steps with a delay of 0.15 seconds between steps.

### EXAMPLE PROGRAM NO. 5 DISCRETE SWEEP ENTRY

In the DISCRETE SWEEP mode the 3326A will step through a sequence of user-entered frequencies on both channels. The sequence can contain up to 63 frequency pairs and an associated dwell time for each pair. Dwell times range from 5 milliseconds to 1000 seconds. Sweeps can be either continuous or single, and reset at the end of the sweep in either case. The discrete sweep elements can be entered from the front panel or through HP-IB under program control. This program demonstrates the use of an array as one of the most convenient ways of entering these elements with a computer.

In this program the array is structured in a way that makes the function of each array element easy to interpret.

RESET the 9836A, SCRATCH the memory, and press EDIT and EXECUTE to enter the following program:

```

10  ! DISCRETE SWEEP ENTRY
20  OPTION BASE 1
30  DIM Discr(5,3)
40  !
50  Source=718
60  !
70  OUTPUT Source;"RST, DCLR, CAL"
80  !
90  ! DISCRETE DATA
100 DATA 5
110 DATA 1000,2000,0.4
120 DATA 3600,1200,0.35
130 DATA 2000,4000,0.3
140 DATA 6000,2000,0.45
150 DATA 5000,10000,0.5
160 !

```

```

170 RESTORE 100
180 READ Num_e1
190 FOR I=1 TO Num_e1
200   READ Discr(I,1),Discr(I,2),Discr(I,3)
210 NEXT I
220 !
230 ! PRINT THE ARRAY
240 PRINT CHR$(12)
250 PRINT "CH A FREQ.      CH B FREQ.      DWELL TIME"
260 IMAGE 9D,5X,9D,5X,5D,3D
270 FOR I=1 TO Num_e1
280   PRINT USING 260;Discr(I,1),Discr(I,2),Discr(I,3)
290 NEXT I
300 !
310 ! SEND ARRAY TO 3326A AS DISCRETE PARAMETERS
320 IMAGE "DSAV ",ZZ
330 FOR I=1 TO Num_e1
340   OUTPUT Source;"CHA, FR ",Discr(I,1)," HZ"
350   OUTPUT Source;"CHB, FR ",Discr(I,2)," HZ"
360   OUTPUT Source;"STIM ",Discr(I,3)," SEC"
370   OUTPUT Source USING 320;I-1
380 NEXT I
390 !
400 ! SET UP 3326A TO PERFORM DISCRETE SWEEP
410 OUTPUT Source;"CHA, FCNA SQR, AM 0 DBU"
420 OUTPUT Source;"CHB, FCNB SIN, AM 3 DBU"
430 OUTPUT Source;"SM DSCR"
440 !
450 ! INITIATE CONTINUOUS DISCRETE SWEEP
460 OUTPUT Source;"SC"
470 END

```

### PROGRAM 5—EXPLANATION AND OPERATION

After the program has been entered, RESET the 9836A and press the RUN key to begin execution of the program.

Set up the oscilloscope with channel 1 and 2 sensitivity of 1 Volt/div. and 0.2 ms/div.

**Line 20** specifies the default lower bound of any arrays to follow. **Line 30** reserves space in memory for the array of discrete sweep parameters to follow. **Line 70** presets the 3326A, clears any previous discrete sweep parameters, and calibrates for optimum accuracy.

**Lines 100-150** construct a data stream in memory that will be used to fill the array of discrete sweep parameters. The array has 5 rows and three columns. Each row contains two frequencies for channels A and B, respectively, and an associated dwell time in seconds. The first data statement (**line 100**) contains the number of rows in the array.

**Line 170** causes the succeeding read statements to begin reading data at **line 100**. This is a precaution, in case there are any other data statements in the program.

**Line 180** reads the first item from the data stream, a count of the rows in the array. This count is used to construct a loop in **lines 190-210**. Each time this loop is executed, it fills a new row in the array, containing channel A and B frequencies and a dwell time.

**Line 240** clears the 9836A display and **line 250** prints a header for the discrete sweep data to follow. The loop beginning in **line 270** is executed once to print each row of the array using the image (format) statement in **line 260**.



Lines 340-390 create a loop that assigns elements in the array to appropriate variables and sends them to the 3326A as discrete sweep parameters.

Line 440 sets the 3326A to the DISCRETE SWEEP mode and line 470 initiates a continuous discrete sweep.

## EXAMPLE PROGRAM NO. 6 INTERROGATING ENTRY PARAMETERS

Virtually all of the current operating parameters of the 3326A can be interrogated by a computer over HP-IB. Any function that has a numeric value associated with it can be interrogated, even if the function is not currently active.

The following program demonstrates the capability of this interrogate function. User-entered frequency, amplitude, and offset parameters are transferred from the 3326A to the computer and displayed.

RESET the 9836A, SCRATCH the memory, and press EDIT and EXECUTE to enter the following program:

```

10      ! INTERROGATING ENTRY PARAMETERS
20      !
30      Source=718
40      !
50      LOCAL Source
60      PRINT CHR$(12)
70      PRINT "THE 3326A IS UNDER FRONT PANEL CONTROL"
80      PRINT "SET UP A FRONT PANEL STATE (FREQUENCY,"
90      PRINT "AMPLITUDE, OFFSET) FOR BOTH CHANNELS"
100     PRINT "AND PRESS CONTINUE"
110     PAUSE
120     !
130     ! INTERROGATE CHANNEL A PARAMETERS
140     OUTPUT Source;"CHA FR?"
150     ENTER Source;Cha_fr$
160     OUTPUT Source;"CHA AM?"
170     ENTER Source;Cha_am$
180     OUTPUT Source;"CHA OF?"
190     ENTER Source;Cha_of$
200     !
210     ! INTERROGATE CHANNEL B PARAMETERS
220     OUTPUT Source;"CHB FR?"
230     ENTER Source;Chb_fr$
240     OUTPUT Source;"CHB AM?"
250     ENTER Source;Chb_am$
260     OUTPUT Source;"CHB OF?"
270     ENTER Source;Chb_of$
280     !
290     ! PRINT PARAMETERS
300     PRINT CHR$(12)
310     PRINT "      CHANNEL A          CHANNEL B:"
320     PRINT
330     IMAGE K,4X,K
340     PRINT USING 330;Cha_fr$,Chb_fr$
350     PRINT USING 330;Cha_am$,Chb_am$
360     PRINT USING 330;Cha_of$,Chb_of$
370     LOCAL Source
380     END

```

## PROGRAM 6—EXPLANATION AND OPERATION

After the program has been entered, RESET the 9836A and press the RUN key to begin execution of the program.

Line 50 insures that the 3326A is in the LOCAL mode so that the user can set up an instrument state from the front panel. Lines 70-100 print a message to the user. Program execution is halted so that 3326A parameters may be entered.

As suggested by the instructions, enter a frequency, amplitude, and offset for each channel from the front panel of the 3326A.

Lines 140-270 interrogate the 3326A for three major parameters from each channel—frequency, amplitude, and offset. The 3326A is directed to output an entry parameter by selecting a channel and supplying an appropriate prefix followed by a question mark.

An ENTER command is used to receive the data and assign it to a string variable. String variables preserve any prefixes or suffixes that the 3326A may send with the requested parameter value. An ENTER command must immediately follow each interrogate command to receive the instrument's output.

The 3326A always responds with Hertz for frequency values, Volts peak-to-peak for amplitude values, seconds for time values, Volts DC for offset values, degrees for phase and percent for duty cycle. Response for internal modulation level is percent or degrees for AM and PM, respectively. In this example, the prefixes, suffixes, and leading and trailing zeroes output by the 3326A are all printed. If string variables were not used to enter the data, other print formats could be used to print only the numeric data.

Lines 310-360 print the requested information on the 9836A CRT. The "K,4X,K" format in line 330 prints each complete parameter string without leading or trailing blanks and prints four spaces between the two strings on each line.

## EXAMPLE PROGRAM NO. 7 SAVING AND RESTORING A 3326A SETUP

It is often helpful to be able to save a specific instrument setup state or states that will be used later in a test procedure. A state can be saved in one of the 3326A's nine internal registers or it can be transferred to the memory of the 9836A for long term storage.

There are two ways to use the computer to store instrument state data. The individual parameters can be interrogated, (as in the previous example program) or the 3326A's LEARN mode may be used to represent the complete instrument state as a compacted string of binary bytes.

Though individual instrument parameters cannot be decoded from the LEARN string, it is a faster and more compact way to represent entire instrument states.

Upon receipt of the LRN command, the 3326A outputs a string of 172 bytes which define an entire setup state. This string can be stored in a computer and later output to the 3326A to restore a setup state.



This program demonstrates the use of the LRN and PRG commands to save and restore a 3326A setup state.

RESET the 9836A, SCRATCH the memory, and press EDIT and EXECUTE to enter the following program:

```

10  ! SAVING OR RESTORING A 3326A SETUP
20  !
30  DIM State$(172)
40  Source=718
50  !
60  LOCAL Source
70  PRINT CHR$(12)
80  PRINT "THE 3326A IS UNDER FRONT PANEL CONTROL"
90  PRINT "SET UP A FRONT PANEL STATE"
100 PRINT "AND PRESS CONTINUE"
110 PAUSE
120 ! RETRIEVE CURRENT INSTRUMENT STATE
130 OUTPUT Source;"SAV 9"
140 OUTPUT Source;"LRN 9"
150 ENTER Source USING "#,172A";State$
160 LOCAL Source
170 PRINT CHR$(12)
180 DISP "CHANGE THE FRONT PANEL STATE"
190 PAUSE
200 ! RESTORE THE CURRENT INSTRUMENT STATE
210 OUTPUT Source;"PRG 9"&State$
220 OUTPUT Source;"RCL 9"
230 LOCAL Source
240 DISP "INSTRUMENT STATE RESTORED"
250 END

```

## PROGRAM 7—EXPLANATION AND OPERATION

After the program has been entered, RESET the 9836A and press the RUN key to begin execution of the program.

**Line 30** reserves space for the string State\$. This string holds the binary characters representing the setup state.

**Lines 60-100** place the 3326A in the LOCAL mode and print a message to set up the 3326A. This state will be stored in the string "State\$". Program execution is then paused.

Follow the instructions and set up a distinctive front panel state.

Press CONTINUE. **Line 130** causes the setup state to be stored in the 3326A's internal register 9. The LRN and PRG commands actually operate with the 3326A's **internal storage registers** rather than the current setup state. Once the state is copied in register 9, **lines 140-150** send the LRN command to retrieve register 9 and read in the data. The format USING "#,172A" causes the 9836A to read 172 characters for the string and suppresses the requirement for terminating conditions (such as the EOI bus management line).

Once the setup state has been stored, **Lines 160-180** restore the 3326A to front panel control and instruct the user to change the setup state. At this point the 3326A can also be preset, turned off, or completely cleared.

Press CONTINUE to resume program execution. **Line 210** sends the 3326A a command to place the contents of the string "State\$" in register 9. Then **line 220** recalls register 9 to restore the previous front panel state. Finally, the instru-

ment is restored to LOCAL control and an appropriate message is displayed on the 9836A CRT. At this point the user can verify that the previous setup state has indeed been restored.

In this same fashion, multiple setup states could be stored in the 9836A and, if desired, saved on disk for later use.

## EXAMPLE PROGRAM NO. 8 SERVICE REQUESTS AND ERRORS

Certain errors and operating conditions of the 3326A can be detected and monitored by the 9836A on an interrupt basis or periodically under program control. Included are both programming and hardware errors and instrument conditions such as the start or completion of a sweep.

When the desired error or condition exists, the user can configure the 3326A to request service from the computer by initiating a Service Request. (SRQ). The computer can detect whether an SRQ has taken place on the bus by analyzing bit 1 (LSB is bit 0) of its interrupt status register (register 4 on the 9836A built in HP-IB interface).

Two methods can be used to analyze the interrupt status of the 9836A HP-IB interface: The program can periodically read the computer's interrupt status register, or it can enable bit 1 of the interrupt enable mask (register 5) to interrupt program execution when an SRQ occurs and bit 1 is set.

In either case, if more than one instrument is on the bus, the computer must conduct a serial poll of all instruments to determine which device requested service. This is done using the SPOLL command and sequentially analyzing the status byte of each instrument that might have generated an SRQ. Under the IEEE-488 definition, the instrument that requires service must have bit 6 of its status byte set.

Once it is determined the 3326A has requested service, the computer can decode the contents of the status byte or, if appropriate, interrogate the error register. A complete description of the status byte and of the error codes are appendices C and B, respectively, of this note. More simply, however, the computer can configure the 3326A to issue an SRQ only when a specific set of errors or operating conditions exist. This set of conditions is determined by a numeric value generated by summing the decimal values of each bit to be checked in the status byte. This value is then sent to the 3326A using the MASK command.

The following program demonstrates the use of the SRQ interrupt in the 9836A and the interrupt mask in the 3326A. In addition, the program interrogates the 3326A for the number of an error generated and prints an appropriate message to the user on the 9836A CRT.



RESET the 9836A, SCRATCH the memory, and press EDIT and EXECUTE to enter the following program:

```

10      ! SERVICE REQUESTS AND ERRORS
20      !
30      Source=718
40      !
50      ABORT 7
60      CLEAR Source
70      OUTPUT Source;"MASK 32 PC"
80      ENABLE INTR 7;2
90      ON INTR 7 GOSUB Srq
100     !
110     INPUT "ENTER CHANNEL A VOLTAGE IN VOLTS",Level
120     OUTPUT Source;"CHA, AM ";Level;" VQ"
130     WAIT .1
140     GOTO 110
150     !
160     Srq: BEEP
170     Status=SPOLL(Source)
180     PRINT CHR$(12)
190     PRINT "E R R O R   D E T E C T E D"
200     OUTPUT Source;"ERR?"
210     ENTER Source;User_error
220     PRINT
230     PRINT "ERROR NUMBER",User_error
240     IF BIT(Status,0) THEN PRINT "PROGRAMMING ERROR"
250     IF BIT(Status,1) THEN PRINT "END OF SWEEP"
260     IF BIT(Status,2) THEN PRINT "SWEEP IN PROGRESS"
270     IF BIT(Status,3) THEN PRINT "HARDWARE ERROR"
280     IF BIT(Status,4) THEN PRINT "READY FOR DATA"
290     IF BIT(Status,7) THEN PRINT "POWER FAILURE/ON"
300     ENABLE INTR 7;2
310     WAIT 3
320     PRINT CHR$(12)
330     RETURN
340     !
350     END

```

## PROGRAM 8—EXPLANATION AND OPERATION

After the program has been entered, RESET the 9836A and press the RUN key to begin execution of the program. No oscilloscope is necessary.

**Line 50** aborts any current activity on the bus and **Line 60** is a selective DEVICE CLEAR command used here to clear the status byte of the 3326A in the event there is an existing SRQ.

**Line 70** sends a Service Request MASK to the 3326A enabling an SRQ only when bit 5 of the 3326A status byte is set. Bit 5 (decimal value 32) is set when an error condition exists in the 3326A. **Line 80** enables program interruption on bit 1 (decimal value 2) of the 9836A status register 5.

**Line 90** directs program execution to the subroutine "SRQ" when an SRQ interrupt is generated.

**Line 110** asks the user to enter a channel A output level. This level is sent to the 3326A as an HP-IB command in **line 120**.

**Line 130** delays program execution long enough to allow the 3326A to generate an error (in the event the entered voltage exceeds limits). Without this delay, the program will be directed to **line 110** before an SRQ can occur. **Line 110** halts program execution to wait for user input and will therefore inhibit interrupt response.

Following the instructions in **line 110**, enter a channel A amplitude in volts and press CONTINUE. Verify that this level is present in the 3326A display. The program will continue to request voltage inputs until an invalid voltage is entered.

When the program transmits an invalid parameter to the 3326A, an SRQ is generated and **line 90** directs program execution to the subroutine "SRQ" beginning on **line 160**. An audible tone is then generated to alert the user.

**Line 170** causes a Serial Poll to be performed and assigns the result to the variable "Status". In the event several instruments are present on the bus and an error can be generated by two or more of them, each instrument must be polled individually. The instrument requesting service will have bit 6 of its status byte set and its status byte can be analyzed to determine the cause of the SRQ.

In this case, it is assumed the SRQ is from the 3326A and the MASK statement in **line 70** insures that the cause is a program or hardware error. The computer, therefore, reads and displays the error in **lines 200-230**.

**Lines 240-290** analyze successive bits of the status byte to determine the condition of the 3326A and print appropriate messages to the user. The interrupt is then re-enabled in **line 300** and the user messages are displayed for three seconds. In **line 320** the 9836A CRT is cleared and the program returns to ask the user for a new voltage level.

To generate messages such as "End Of Sweep" or "Sweep In Progress", simply enter an invalid Channel A voltage, press the LOCAL key on the 3326A, and initiate a single or continuous sweep before pressing the CONTINUE key.



# APPENDIX A

## HP 3326A PROGRAMMING CODES

Mnemonic	Range	Suffix	Front Panel Control	Description Resolution Syntax
ACAL	0-1 or —	— OFF, ON	AUTO	AutoCALibration Syntax: "ACALO" "ACAL OFF"
AEA	0-1 or —	— OFF, ON	CH A	Channel A External Am Syntax: "AEA1" "AEA ON"
AEP	0-1 or —	— OFF, ON	CH A	Channel A External Pm Syntax: "AEP1" "AEP ON"
AIA	0-1 or —	— OFF, ON	CH A	Channel A Internal Am Syntax: "AIA1" "AIA ON"
AIP	0-1 or —	— OFF, ON	CH A	Channel A Internal Pm Syntax: "AIP1" "AIP ON"
AM	0-10 V	VO, VRMS, DBM, DBV	AMPTD	AMplitude Resolution: 1 mVpp Syntax: "AM1.125VO"
BEA	0-1 or —	— OFF, ON	CH B	Channel B External Am Syntax: "BEA1" "BEA ON"
BEP	0-1 or —	— OFF, ON	CH B	Channel B External Pm Syntax: "BEP1" "BEP ON"
BUSM	1-2	—	none	BUS Mode Syntax: "BUSM2"
CAL	—	—	MANUAL	CALibrate Syntax: "CAL"
CF	0-13 MHz	HZ, KHZ, MHZ	CNTR FREQ	Center Frequency Resolution: 1 $\mu$ Hz $f < 100$ kHz $1$ MHz $f \geq 100$ kHz Syntax: "CF10KHZ"
CFM	—	—	MKR->CF	Center Frequency equals Marker value Syntax: "CFM"
CHA	—	—	CHAN	select CHannel A Syntax: "CHA"
CHB	—	—	CHAN	select CHannel B Syntax: "CHB"
CMB	0-1 or —	— OFF, ON	COMBINED	CoMBiner Syntax: "CMB1" "CMB ON"



# HP 3326A PROGRAMMING CODES (CONT'D)

Mnemonic	Range	Suffix	Front Panel Control	Description Resolution Syntax
CMD	1 or —	—  INT	SELECT	Calibration MoDe - INTernal Syntax: "CMD1" "CMD INT"
	2 or —	—  EXT		Calibration MoDe - EXTernal Syntax: "CMD1" "CMD EXT"
	3 or —	—  MULT		Calibration MoDe - MULTiphasE Syntax: "CMD3" "CMD MULT"
COF	—	—	CLR $\phi$ OFS	Clear phase Offset Syntax: "COF"
DBM	—	—	units	DBM
DBV	—	—	units	DBV
DC	—	—	none	suffix DC function output
DCLR	—	—	RST DISCRETE	Discrete sweep CLear Syntax: "DCLR"
DEG	—	—	units	DEGrees
DN	—	—	none	DowN increment by EINC value Syntax: "DN"
DRCL	00-62	—	RCL DISCRETE	Discrete ReCaLI Syntax: "DRCL02"
DSAV	00-62	—	DISCRETE STO	Discrete SAVe Syntax: "DSAV02"
DSCR	—	—		Suffix - DiSCReTE Sweep Mode
DISP	0-1 or —	—  OFF, ON	none	DISPlay control Syntax: "DISP1" "DISP ON"
DUTY	1-99%	PC	DUTY CYCLE	DUTY cycle Resolution: 0.01 % Syntax: "DUTY25.05PC"
EINC	see description		none	Entry INCrement for UP, DN, TUP, and TDN commands Use increment resolution and suffix appropriate for entry value modified Syntax: "EINC1HZ" "EINC.1VRMS"



# HP 3326A PROGRAMMING CODES (CONT'D)

Mnemonic	Range	Suffix	Front Panel Control	Description Resolution Syntax
ERR?	—	—	none	ERRor code Syntax: "ERR?"
EXT	—	—	none	suffix for EXTernal calibration
FCNA	0 or —	—  OFF	CHA	FunCtioN channel A OFF Syntax: "FCNA0" "FCNA OFF"
	1 or —	—  SIN		FunCtioN channel A SINe Syntax: "FCNA1" "FCNA SIN"
	2 or —	—  SQR		FunCtioN channel A SQuaRe Syntax: "FCNA2" "FCNA SQR"
	3 or —	—  DC		FunCtioN channel A DC Syntax: "FCNA3" "FCNA DC"
	0 or —	—  OFF		FunCtioN channel B OFF Syntax: "FCNB0" "FCNB OFF"
FCNB	1 or —	—  SIN	CHB	FunCtioN channel B SINe Syntax: "FCNB1" "FCNB SIN"
	2 or —	—  SQR		FunCtioN channel B SQuaRe Syntax: "FCNB2" "FCNB SQR"
	3 or —	—  DC		FunCtioN channel B DC Syntax: "FCNB3" "FCNB DC"
	0-13 MHz	HZ, KHZ, MHZ	FREQ	FRequency Resolution: 1 $\mu$ Hz $f < 100$ kHz 1 mHz $f \geq 100$ kHz Syntax: "FR7.500003MHZ"
FR	0-1 or —	—  OFF, ON	CH A HV	High Voltage channel A Syntax: "HVA1" "HVA ON"
HVA	0-1 or —	—  OFF, ON	CH B HV	High Voltage channel B Syntax: "HVB1" "HVB ON"
HVB	—	—	units	HertZ



# HP 3326A PROGRAMMING CODES (CONT'D)

Mnemonic	Range	Suffix	Front Panel Control	Description Resolution Syntax
ID?	—	—	none	IDentification Syntax: "ID?"
INT	—	—	none	suffix for INTernal calibration
KHZ	—	—	units	KiloHertz
LRN	0-9	—	none	LeaRN (read) nonvolatile memory Syntax: "LRN3"
MASK	0-255	PC	none	srq MASK (weighted binary sum of bit positions) Syntax: "MASK32PC"
MF	0-13MHz	HZ, KHZ, MHZ	MKR FREQ	Marker Frequency Resolution: 1 $\mu$ Hz $f < 100$ kHz 1 mHz $f \geq 100$ kHz Syntax: "MF8.0MHZ"
MFY	0-1 or —	—  OFF, ON	ON/OFF	front panel ModiFY control Syntax: "MFY1" "MFY ON"
MHZ	—	—	units	MegaHertz
ML	0-100% or 0-360°	PC  DEG	% AM/PM DEV	Modulation Level Resolution: 0.1 % or 1° Syntax: "ML30PC"
MODE	1 or —	—  TWOC	MODE	MODE TWO Channel Syntax: "MODE1" "MODE TWOC"
	2 or —	—  TWOP		MODE TWO Phase Syntax: "MODE2" "MODE TWOP"
	3 or —	—  TWOT		MODE TWO Tone Syntax: "MODE3" "MODE TWOT"
	4 or —	—  PULS		MODE PULSe Syntax: "MODE4" "MODE PULS"
MS	—	—	units	MilliSeconds
MULT	—	—	none	suffix for MULTiphase calibration
NOM	—	—	none	NO Modulation Syntax: "NOM"
OF	+ -5 V	VO	DC OFFSET	Offset Resolution: 10 mV Syntax: "OF3VO"



# HP 3326A PROGRAMMING CODES (CONT'D)

Mnemonic	Range	Suffix	Front Panel Control	Description Resolution Syntax
OFF	—	—	none	suffix to disable function
ON	—	—	none	suffix to enable function
PC	—	—	units	PerCent
PH	+ -720°	DEG	PHASE	PHase Resolution: 0.01° Syntax: "PH180DEG"
PRG	0-9	—	none	PRoGram (restore) nonvolatile memory Syntax: "PRG3"
PULS	—	—	none	suffix for PULSe mode
RAMP	—	—	none	suffix for RAMP sweep
RCL	0-9	—	RECALL	ReCaLI Syntax: "RCL3"
RDY?	—	—	none	ReaDY Syntax: "RDY?"
REV?	—	—	none	REVisiOn Syntax: "REV?"
RST	—	—	INSTR PRESET	ReSeT Syntax: "RST"
SAV	0-9	—	STORE	SAVe Syntax: "SAV3"
SC	—	—	CONT	Sweep, Continuous Syntax: "SC"
SEC	—	—	units	SEConds
SER?	—	—	none	SERial number Syntax: "SER?"
SIN	—	—		suffix for SiNe wave function
SM	1	—	TRIANGLE	Sweep Mode - linear RAMP Syntax: "SM1" "SM RAMP"
	or —	RAMP		
	2 or —	—	TRIANGLE	Sweep Mode - linear TRianGLE Syntax: "SM2" "SM TRGL"
	—	TRGL		
	3 or —	—	DISCRETE	Sweep Mode - DisCRete Syntax: "SM3" "SM DSCR"
	—	DSCR		



# HP 3326A PROGRAMMING CODES (CONT'D)

Mnemonic	Range	Suffix	Front Panel Control	Description Resolution Syntax
SP	0-13 MHz	HZ, KHZ, MHZ	STOP FREQ	StoP frequency Resolution: 1 $\mu$ Hz $f < 100$ kHz 1 mHz $f \geq 100$ kHz Syntax: "SP7.125MHZ"
SPAN	0-13 MHz	HZ, KHZ, MHZ	SPAN	sweep frequency SPAN Resolution: 1 $\mu$ Hz $f < 100$ kHz 1 mHz $f \geq 100$ kHz Syntax: "SPAN10.125MHZ"
SPE	0-1 or —	— OFF, ON	CH A	Synchronous Phase modulation External Syntax: "SPE1" "SPE ON"
SQR	—	—	none	suffix for SQuaRe wave function
SRE	—	—	RESET SWP	Sweep REset Syntax: "SRE"
SS	—	—	SINGLE	Sweep Single Syntax: "SS"
ST	0-13 MHz	HZ, KHZ, MHZ	START FREQ	STart frequency Resolution: 1 $\mu$ Hz $f < 100$ kHz 1 mHz $f \geq 100$ kHz Syntax: "ST3.5KHZ"
STC	—	—	none	Sweep on Trigger - Continuous Syntax: "STC"
STIM	5 ms-1000 s	SEC, MS	TIME	Sweep TIME Resolution: 1 mS Syntax: "STIM.3S"
STS	—	—	none	Sweep on Trigger - Single Syntax: "STS"
TDN	—	—	none	Trigger Down increment by EINC amount Syntax: "TDN"
TOFF	—	—	none	Trigger OFF Syntax: "TOFF"
TRGL	—	—	suffix	TRianGLE sweep mode
TST	—	—	SELF TEST	self TeST Syntax: "TST"
TUP	—	—	none	Trigger UP increment by EINC amount Syntax: "TUP"



## HP 3326A PROGRAMMING CODES (CONT'D)

Mnemonic	Range	Suffix	Front Panel Control	Description Resolution Syntax
TWOC	—	—	none	suffix for TWO Channel mode
TWOP	—	—	none	suffix for TWO Phase mode
TWOT	—	—	none	suffix for TWO Tone mode
UP	—	—	none	UP increment by EINC value Syntax: "UP"
VO	—	—	units	VOLts peak-to-peak for amplitude. VOLts dc for dc offset.
VRMS	—	—	units	Volts RMS for amplitude
WAIT	—	—	none	no operation Syntax: "WAIT"
ZPH	—	—	ASGN ZERO $\phi$	Zero PHase Syntax: "ZPH"

## APPENDIX B

### HP 3326A HP-IB ERROR CODES

#### CODE DESCRIPTION

- |    |   |
|----|---|
| 10 | HP-IB command has syntax error or contains illegal characters   |
| 11 | Front panel key pressed while HP 3326A in remote  |
| 12 | Front panel key pressed while HP 3326A in local lockout   |
| 20 | Value entered for selected parameter exceeds valid limits   |
| 21 | In 2 TONE mode, channel B offset frequency greater than 100 kHz   |
| 22 | Amplitude and dc offset values incompatible   |
| 23 | Discrete frequency sweep element save nonsequential with existing elements, or instrument state save breaks continuity of discrete frequency elements |
| 24 | Marker frequency entered is outside sweep span  |
| 25 | Frequency value greater than 1 MHz entered with high voltage option active  |



## APPENDIX B. HP 3326A HP-IB ERROR CODES (CONT.)

### CODE DESCRIPTION

26	Frequency value greater than 5 kHz entered with internal PM active, or greater than 100 kHz with internal AM active
29	Combiner is enabled, and nonzero dc offset entered with function other than DC only, or amplitude value greater than half the normal limits
30	In 2 TONE mode with channel B high voltage option enabled, channel B frequency cannot track change to channel A frequency
40	Value that cannot be displayed has been interrogated over the HP-IB
46	Internal modulation enabled and Channel B amplitude or offset selected as display value
47	Channel B phase selected as display value when PULSE mode enabled
50	Units conversion results in zero display value
60	Units key selected improper for parameter selected
65	High voltage option enabled and dBm selected as units
70	Increment value or units incompatible with displayed value
80	Combiner selected but not enabled because current amplitude value is too large
86	Combiner selected but not enabled because Internal AM or PM is enabled
87	1) In PULSE mode—sine wave output selected, combiner selected, or zero phase assigned to channel B 2) In 2 CHANNEL, 2 TONE, or PULSE mode—channel B phase offset cleared 3) In 2 TONE, 2 PHASE, or PULSE mode—internal AM or PM selected 4) In 2 CHANNEL mode—synchronous PM selected
88	Internal PM selected with channel B frequency greater than 5 kHz, or internal AM selected with channel B frequency greater than 100 kHz
89	Combiner selected but not enabled because AM or PM enabled
90	Frequency sweep start and stop frequencies are equal for both channels
94	Pulse duty cycle too narrow for sweep range
95	High voltage option enabled and sweep frequency is greater than 1 MHz
96	Channel B frequency exceeds 5 kHz internal PM limit or 100 kHz internal AM limit during sweep
100	Sweep rate less than 5 mHz per second or greater than 0.5 MHz per second
110	No discrete frequency sweep elements exist for discrete frequency sweep
114	Frequency too high for duty cycle requested during discrete frequency sweep
115	High voltage option enabled and discrete frequency sweep element frequency exceeds 1 MHz
116	Channel B frequency exceeds the 5 kHz internal PM limit or 100 kHz internal AM limit during discrete frequency sweep
117	Discrete frequency elements in memory incompatible with selected mode
120	Cannot clear channel A phase offset
130	High voltage option selected and not installed
136	Channel B high voltage option selected with internal modulation
138	High voltage option selected when frequency is greater than 1 MHz
140	A checksum error for recall, learn, or program operation
150	Current instrument configuration incompatible with recalled or programmed state
160	An error is detected in an instrument state recalled from memory and instrument state is replaced with preset state
170	Channel A output is overloaded
171	Channel B output is overloaded
172	SYNC output is overloaded
173	Channel A voltage controlled oscillator is unlocked
180	HP 3326A cannot lock to external reference signal that is present
190	Unsuccessful internal AM or PM calibration
191	Unsuccessful phase calibration
192	Unsuccessful amplitude calibration
193	Unsuccessful dc offset calibration
194	Unsuccessful residual dc offset calibration



## APPENDIX C

### HP 3326A STATUS BYTE

BIT NUMBER	DECIMAL VALUE	DESCRIPTION
B7	128	<b>POWER RESTORED.</b> Set when power is restored to the HP 3326A after power is interrupted. Reset when the HP 3326A is preset or receives a device clear, selected device clear, or RST command.
B6	64	<b>REQUIRE SERVICE.</b> Set when the HP 3326A requires service (sent an SRQ). Cleared along with the SRQ line when a serial poll is performed. It is also cleared when the condition causing the SRQ is removed.
B5	32	<b>ERROR.</b> Set when either a program or hardware error condition exists for the HP 3326A. Reset when the HP 3326A is preset, or receives a device clear command, selected device clear command, RST command, or when the error register is read with the IERR or ERR? HP-IB command.
B4	16	<b>READY.</b> Set when the HP 3326A has executed the last HP-IB command and is ready for the next command. Reset when the HP 3326A receives a device dependent command, device clear command, selected device clear command, or trigger.

BIT NUMBER	DECIMAL VALUE	DESCRIPTION
B3	8	<b>HARDWARE ERROR.</b> Set when the HP 3326A detects an internal failure. Reset with an INSTR PRESET, device clear command, selected device clear command, RST command, or when the error register is read with the IERR or ERR? HP-IB command.
B2	4	<b>SWEEP START/IN PROGRESS.</b> Set when the HP 3326A starts a sweep. Reset when the sweep is stopped (either by reaching the stop frequency or aborted by a front panel or HP-IB command). It is also reset when the HP 3326A is preset or receives a device clear command, selected device clear command, or RST command.
B1	2	<b>SWEEP STOPPED.</b> Set when the HP 3326A ends a sweep normally. Reset when the HP 3326A is preset or receives a device clear command, selected device clear command, or RST command.
B0	1	<b>PROGRAM ERROR.</b> Set when the HP 3326A receives an invalid HP-IB command (e.g. command syntax or incompatible command for mode selected). Reset when the HP 3326A is preset or receives a device clear command, selected device clear command, or RST command.





**FOR MORE INFORMATION, CALL YOUR LOCAL HP SALES OR SERVICE OFFICE** or East (201) 265-5000 • Midwest (312) 255-9800 • South (404) 955-1500 • West (213) 970-7500 or (415) 968-9200 **OR WRITE**, Hewlett-Packard, 1820 Embarcadero, Palo Alto, California 94303. **IN EUROPE, CALL YOUR LOCAL HP SALES or SERVICE OFFICE OR WRITE**, Hewlett-Packard S.A., 7, rue du Bois-du-Lan Case Postale 365 CH 1217 Meyrin 1 - Geneva, Switzerland. **IN JAPAN**, Yokogawa-Hewlett-Packard Ltd., 1-27-15, Yabe Sagami-hara City, Kanagawa Prefecture, Japan 229.

5953-5135

Printed in U.S.A.